// File: contracts/PLSBLockBulk.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.18;

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

interface iPLSB {

  function approve(address spender, uint amount) external returns (bool);
  function transfer(address to, uint amount) external returns (bool);
  function transferFrom(address sender, address recipient, uint amount) external returns(bool);

}


interface iPLSBLock {

  function lockUpAndMint(
    uint lockupAmount,
    uint tipAmount,
    uint _lockTimeInterval,
    bool transferable,
    address mintTo,
    string memory uri
  ) external;

}


contract PLSBLockBulk is ReentrancyGuard {
  iPLSBLock public plsbLock;
  iPLSB public plsb;

  address public txnFeeSendTo;
  uint public baseTxnFee;

  error InputsDoNotMatch(uint lockupAmtsLength, uint lockTimesLength);

  constructor(address _plsb, address _plsbLock, address _txnFeeSendTo) {
    plsbLock = iPLSBLock(_plsbLock);
    plsb = iPLSB(_plsb);

    txnFeeSendTo = _txnFeeSendTo;
    baseTxnFee = 0.001 ether;

    plsb.approve(address(plsbLock), type(uint).max);
  }

  receive() payable external {
    payable(txnFeeSendTo).transfer(msg.value);
  }

  function _txnFee(uint numToCreate) internal view returns(uint) {
    uint txnFee = baseTxnFee * numToCreate;
    return txnFee;
  }

  function bulkLockUpAndMint(
    uint[] calldata lockupAmounts,
    uint[] calldata lockTimeIntervals,
    bool transferable,
    address mintTo,
    string[] calldata uris
  ) public payable nonReentrant {
    uint loopLength = lockupAmounts.length;
    uint txnFee = _txnFee(loopLength);

    if(loopLength != lockTimeIntervals.length) {
      revert InputsDoNotMatch(loopLength, lockTimeIntervals.length);
    }
    if(loopLength != uris.length) {
      revert InputsDoNotMatch(loopLength, uris.length);
    }

    if(msg.value != txnFee) {
      revert("Txn Fee invalid");
    }

    payable(txnFeeSendTo).transfer(txnFee);

    uint totalLockupAmount;
    for(uint i; i < loopLength;) {
      totalLockupAmount += lockupAmounts[i];

      unchecked {
        i++;
      }
    }

    plsb.transferFrom(msg.sender, address(this), totalLockupAmount);

    for(uint i; i < loopLength;) {

      plsbLock.lockUpAndMint(
        lockupAmounts[i],
        0,
        lockTimeIntervals[i],
        transferable,
        mintTo,
        uris[i]
      );

      unchecked {
        i++;
      }
    }

  }

}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}